
Understanding and Using
Windows API Calls
In the Programming with the Windows API chapter of our Excel 2002 VBA
Programmers Reference, we approached the subject of using Windows API
calls by explaining how to locate the definitions for various functions on the
MSDN Web site and translate those functions for use in VBA. The idea
was to enable readers to browse through the API documentation and use
anything of interest they found.
In reality, extremely few people use Windows API calls in that manner;
indeed, trying to include previously unexplored API calls in our Excel
applications is very likely to result in a maintenance problem, because it’s
doubtful that another developer will understand what we were trying to
do. Instead, most of us go to Google and search the Web or the newsgroups for the answer to a problem and find that the solution requires the
use of API calls. (Searching Google for “Excel Windows API” results in
more than 200,000 Web pages and 19,000 newsgroup posts.) We copy the
solution into our application and hope it works, usually without really
understanding what it does. This chapter shines a light on many of those
solutions, explaining how they work, what they use the API calls for, and
how they can be modified to better fit our applications. Along the way, we
fill in some of the conceptual framework of common Windows API techniques and terminology.
By the end of the chapter, you will be comfortable about including API
calls in your applications, understand how they work, accept their use in
the example applications we develop in this book and be able to modify
them to suit your needs.

Overview
When developing Excel-based applications, we can get most things done
by using the Excel object model. Occasionally, though, we need some

information or feature that Excel doesn’t provide. In those cases, we can
usually go directly to the files that comprise the Windows operating system
to find what we’re looking for. The first step in doing that is to tell VBA the
function exists, where to find it, what arguments it takes and what data type
it returns. This is done using the Declare statement, such as that for
GetSystemMetrics:
Declare Function GetSystemMetrics Lib "user32" _
(ByVal nIndex As Long) As Long

This statement tells the VBA interpreter that there is a function called
GetSystemMetrics located in the file user32.exe (or user32.dll, it’ll check
both) that takes one argument of a Long value and returns a Long value.
Once defined, we can call GetSystemMetrics in exactly the same way as if
it is the VBA function:
Function GetSystemMetrics(ByVal nIndex As Long) As Long
End Function

The Declare statements can be used in any type of code module, can
be Public or Private (just like standard procedures), but must always be
placed in the Declarations section at the top of the module.

Finding Documentation
All of the functions in the Windows API are fully documented in the
Windows Development/Platform SDK section of the MSDN library on the
Microsoft Web site, at http://msdn.microsoft.com/library,
although the terminology used and the code samples tend to be targeted
at the C++ developer. A Google search will usually locate documentation
more appropriate for the Visual Basic and VBA developer, but is unlikely
to be as complete as MSDN. If you’re using API calls found on a Web site,
the Web page will hopefully explain what they do, but it is a good idea to
always check the official documentation for the functions to see whether
any limitations or other remarks may affect your usage.
Unfortunately, the MSDN library’s search engine is significantly worse
than using Google to search the MSDN site. We find that Google always
gives us more relevant pages than MSDN’s search engine. To use Google
to search MSDN, browse to http://www.google.com and click the
Advanced Search link. Type in the search criteria and then in the Domain
edit box type msdn.microsoft.com to restrict the search to MSDN.

Finding Declarations
It is not uncommon to encounter code snippets on the Internet that
include incorrect declarations for API functions—such as declaring an
argument’s data type as Integer or Boolean when it should be Long.
Although using the declaration included in the snippet will probably work
(hopefully the author tested it), it might not work for the full range of possible arguments that the function accepts and in rare cases may cause
memory corruption and data loss. The official VBA-friendly declarations
for many of the more commonly used API functions can be found in the
win32api.txt file, which is included with a viewer in the Developer Editions
of Office 97–2002, Visual Basic 6 and is available for download from
http://support.microsoft.com/?kbid=178020. You’ll notice from
the download page that the file hasn’t been updated for some time. It
therefore doesn’t include the declarations and constants added in recent
versions of Windows. If you’re using one of those newer declarations, you’ll
have to trust the Web page author, examine a number of Web pages to
check that they all use the same declaration or create your own VBAfriendly declaration by following the steps we described in the Excel 2002
VBA Programmers Reference.

Finding the Values of Constants
Most API functions are passed constants to modify their behavior or specify the type of value to return. For example, the GetSystemMetrics
function shown previously accepts a parameter to specify which metric
we want, such as SM_CXSCREEN to get the width of the screen in pixels
or SM_CYSCREEN to get the height. All of the appropriate constants
are shown on the MSDN page for that declaration. For example,
the GetSystemMetrics function is documented at http://
msdn.microsoft.com/library/en-us/sysinfo/base/
getsystemmetrics.asp and shows more than 70 valid constants.

Although many of the constants are included in the win32api.txt
file mentioned earlier, it does not include constants added for
recent versions of Windows. The best way to find these values is by downloading and installing the core Platform SDK from http://
www.microsoft.com/msdownload/platformsdk/
sdkupdate/. This includes all the C++ header files that were used to

build the DLLs, in a subdirectory called \include. The files in this directory can be searched using normal Windows file searching to find the file that

contains the constant we’re interested in. For example, searching for
SM_CXSCREEN gives the file winuser.h. Opening that file and searching
within it gives the following lines:
#define SM_CXSCREEN
#define SM_CYSCREEN

0
1

These constants can then be included in your VBA module by declaring them as Long variables with the values shown:
Const SM_CXSCREEN As Long = 0
Const SM_CYSCREEN As Long = 1

Sometimes, the values will be shown in hexadecimal form, such as
0x8000, which can be converted to VBA by replacing the 0x with &h and
adding a further & on the end, such that
#define KF_UP

0x8000

becomes
Const KF_UP As Long = &h8000&

Understanding Handles
Within VBA, we’re used to setting a variable to reference an object using
code like
Set wkbBackDrop = Workbooks("Backdrop.xls")

and releasing that reference by setting the variable to Nothing (or letting
VBA do that for us when it goes out of scope at the end of the procedure).
Under the covers, the thing that we see as the Backdrop.xls workbook is
just an area of memory containing data structured in a specific way that
only Excel understands. When we set the variable equal to that object, it is
just given the memory location of that data structure. The Windows operating system works in a very similar way, but at a much more granular level;
almost everything within Windows is maintained as a small data structure
somewhere. If we want to work with the item that is represented by that
structure (such as a window), we need to get a reference to it and pass that

reference to the appropriate API function. These references are known as
handles and are just ID numbers that Windows uses to identify the data
structure. Variables used to store handles are usually given the prefix h and
are declared As Long.
When we ask for the handle to an item, some functions—such as
FindWindow—give us the handle to a shared data structure; there is only
one data structure for each window, so every call to FindWindow with the
same parameters will return the same handle. In these cases, we can just
discard the handle when we’re finished with it. In most situations, however, Windows allocates an area of memory, creates a new data structure for
us to use and returns the handle to that structure. In these cases, we must
tidy up after ourselves, by explicitly telling Windows that we’ve finished
using the handle (and by implication, the memory used to store the data
structure that the handle points to). If we fail to tidy up correctly, each call
to our routine will use another bit of memory until Windows crashes—this
is known as a memory leak. The most common cause of memory leaks is
forgetting to include tidy-up code within a routine’s error handler. The
MSDN documentation will tell you whether you need to release the handle and which function to call to do it.

Encapsulating API Calls
GetSystemMetrics is one of the few API calls that can easily be used in
isolation—it has a meaningful name, takes a single parameter, returns a
simple result and doesn’t require any preparation or cleanup. So long as
you can remember what SM_CXSCREEN is asking for, it’s extremely easy
to call this function; GetSystemMetrics(SM_CXSCREEN) gives us the
width of the screen in pixels.
In general practice, however, it is a very good idea to wrap your API
calls inside their own VBA functions and to place those functions in modules dedicated to specific areas of the Windows API, for the following
reasons:
■

■

The VBA routine can include some validity checks before trying to
call the API function. Passing invalid data to API functions will often
result in a crash.
Most of the textual API functions require string variables to be
defined and passed in, which are then populated by the API function. Using a VBA routine hides that complexity.

Many API functions accept parameters that we don’t need to use. A
VBA routine can expose only the parameters that are applicable to
our application.
Few API functions can be used in isolation; most require extra
preparatory and clean up calls. Using a VBA routine hides that
complexity.
The API declarations themselves can be declared Private to the
module in which they’re contained, so they can be hidden from use
by other developers who may not understand how to use them; their
functionality can then be exposed through more friendly VBA
routines.
Some API functions, such as the encryption or Internet functions,
require an initial set of preparatory calls to open resources, a number of routines that use those resources and a final set of routines
to close the resources and tidy up. Such routines are ideally encapsulated in a class module, with the Class_Initialize and
Class_Terminate procedures used to ensure the resources are
opened and closed correctly.
By using dedicated modules for specific areas of the Windows API,
we can easily copy the routines between applications, in the knowledge that they are self-contained.

When you start to include lots of API calls in your application, it quickly becomes difficult to keep track of which constants belong to which functions. We can make the constants much easier to manage if we encapsulate
them in an enumeration and use that enumeration for our VBA function’s
parameter, as shown in Listing 9-1. By doing this, the applicable constants
are shown in the Intellisense list when the VBA function is used, as shown
in Figure 9-1. The ability to define enumerations was added in Excel 2000.
Listing 9-1 Encapsulating the GetSystemMetrics API Function and Related Constants
'Declare all the API-specific items Private to the module
Private Declare Function GetSystemMetrics Lib "user32" _
(ByVal nIndex As Long) As Long
Private Const SM_CXSCREEN As Long = 0
Private Const SM_CYSCREEN As Long = 1
'Wrap the API constants in a public enumeration,

'so they appear in the Intellisense dropdown
Public Enum SystemMetricsConstants
smScreenWidth = SM_CXSCREEN
smScreenHeight = SM_CYSCREEN
End Enum
'Wrapper for the GetSystemMetrics API function,
'using the SystemMetricsConstants enumeration
Public Function SystemMetrics( _
ByVal uIndex As SystemMetricsConstants) As Long
SystemMetrics = GetSystemMetrics(uIndex)
End Function

Figure 9-1 By Using the Enumeration, the Relevant Constants Appear in the
Intellisense Drop-Down

Working with the Screen
The procedures included in this section all relate to the Windows screen
and can be found in the MScreen module of the API Examples.xls
workbook.

Reading the Screen Resolution
The GetSystemMetrics API function has been used to illustrate the general concepts above. It can be used to discover many of the simpler aspects
of the operating system, from whether a mouse or network is present to the
height of the standard window title bar. By far its most common use in
Excel is to find the screen resolution, to check that it is at least a minimum
size (for example, 800×600) or to work out which userform to display if you
have different layouts optimized for different resolutions. The code in
Listing 9-2 wraps the GetSystemMetrics API function, exposing it as separate ScreenWidth and ScreenHeight functions.

Listing 9-2 Reading the Screen Resolution
'Declare all the API-specific items Private to the module
Private Declare Function GetSystemMetrics Lib "user32" _
(ByVal nIndex As Long) As Long
Private Const SM_CXSCREEN = 0
'Screen width
Private Const SM_CYSCREEN = 1
'Screen height
'The width of the screen, in pixels
Public Function ScreenWidth() As Long
ScreenWidth = GetSystemMetrics(SM_CXSCREEN)
End Function
'The height of the screen, in pixels
Public Function ScreenHeight() As Long
ScreenHeight = GetSystemMetrics(SM_CYSCREEN)
End Function

Finding the Size of a Pixel
In general, Excel measures distances in points, whereas most API functions use pixels and many ActiveX controls (such as the Microsoft Flexgrid)
use twips. A point is defined as being 1/72 (logical) inches, and a twip is
defined as 1/20th of a point. To convert between pixels and points, we need
to know how many pixels Windows is displaying for each logical inch. This
is the DPI (dots per inch) set by the user in Control Panel > Display >
Settings > Advanced > General > Display, which is usually set at either
Normal size (96 DPI) or Large size (120 DPI). In versions of Windows
prior to XP, this was known as Small Fonts and Large Fonts. The value of
this setting can be found using the GetDeviceCaps API function, which is
used to examine the detailed capabilities of a specific graphical device,
such as a screen or printer.
Device Contexts
One of the fundamental features of Windows is that applications can interact with all graphical devices (screens, printers, or even individual picture
files) in a standard way. This is achieved by operating through a layer of
indirection called a device context, which represents a drawing layer. An
application obtains a reference (handle) to the drawing layer for a specific
device (for example, the screen), examines its capabilities (such as the size

of a dot, whether it can draw curves and how many colors it supports),
draws onto the drawing layer and then releases the reference. Windows
takes care of exactly how the drawing layer is represented on the graphical
device. In this example, we’re only examining the screen’s capabilities.
The code to retrieve the size of a pixel is shown in Listing 9-3.
Remember that when adding this code to an existing module, the declarations must always be placed at the top of the module.
Listing 9-3 Finding the Size of a Pixel
Private Declare Function GetDC Lib "user32" _
(ByVal hwnd As Long) As Long
Private Declare Function GetDeviceCaps Lib "gdi32" _
(ByVal hDC As Long, ByVal nIndex As Long) As Long
Private Declare Function ReleaseDC Lib "user32" _
(ByVal hwnd As Long, ByVal hDC As Long) As Long
Private Const LOGPIXELSX = 88

'Pixels/inch in X

'A point is defined as 1/72 inches
Private Const POINTS_PER_INCH As Long = 72
'The size of a pixel, in points
Public Function PointsPerPixel() As Double
Dim hDC As Long
Dim lDotsPerInch As Long
hDC = GetDC(0)
lDotsPerInch = GetDeviceCaps(hDC, LOGPIXELSX)
PointsPerPixel = POINTS_PER_INCH / lDotsPerInch
ReleaseDC 0, hDC
End Function

The first thing to notice about this routine is that we cannot just call
GetDeviceCaps directly; we need to give it a handle to the screen’s device
context. This handle is obtained by calling the GetDC function, where the
zero parameter conveniently gives us the device context for the screen. We
then call GetDeviceCaps, passing the constant LOGPIXELSX, which asks

for the number of pixels per logical inch horizontally. (For screens, the horizontal and vertical DPI is the same, but it might not be for printers, which
is why circles on screen often print out as ovals.) With Normal size chosen,
we get 96 dots per inch. We divide the 72 points per inch by the 96 DPI,
telling us that a dot (that is, pixel) is 0.75 points; so if we want to move
something in Excel by one pixel, we need to change its Top or Left by 0.75.
With Large Size selected, a pixel is 0.6 points.
Every time we use GetDC to obtain a handle to a device context, we
use up a small amount of Window’s graphical resources. If we didn’t
release the handle after using it, we would eventually use up all of
Window’s graphical resources and crash. To avoid that, we have to be sure
to release any resources we obtain, in this case by calling ReleaseDC.

Working with Windows
Everything that we see on the screen is either a window or is contained
within a window, from the Windows desktop to the smallest popup tooltip.
Consequently, if we want to modify something on the screen, we always
start by locating its window. The windows are organized into a hierarchy,
with the desktop at the root. The next level down includes the main windows for all open applications and numerous system-related windows.
Each application then owns and maintains its own hierarchy of windows.
Every window is identified by its window handle, commonly referred to as
hWnd. By far the best tool for locating and examining windows is the
Spy++ utility that is included with Visual Studio. Figure 9-2 shows the
Spy++ display for the window hierarchy of a typical Excel session.
Window Classes
As well as showing the hierarchy, the Spy++ display shows three key
attributes for each window: the handle (in hexadecimal), the caption and
the class. Just like class modules, a window class defines a type of window.
Some classes, such as the ComboBox class, are provided by the Windows
operating system, but most are defined as part of an application. Each
window class is usually associated with a specific part of an application,
such as XLMAIN being Excel’s main application window. Table 9-1 lists
the window classes shown in the Spy++ hierarchy and their uses, plus
some other window classes commonly encountered during Excel application development.

Figure 9-2 The Spy++ Display of the Excel Window Hierarchy
Table 9-1 Excel Window Classes and Their Uses
Window Class

Usage

XLMAIN
EXCEL;

The main Excel application window.
The left half of the formula bar, including the Name
drop-down.
A standard Windows combo box (in this case, it’s the
Name drop-down).
The edit box section of the formula bar.
The four command bar docking areas (top, left,
right and bottom).
A command bar.
The Excel desktop.
A workbook window. In this example, Book1 has
two windows open.
A window used to provide in-sheet editing of
embedded charts.
The status bar.

ComboBox
EXCEL<
EXCEL2
MsoCommandBar
XLDESK
EXCEL7
EXCELE
EXCEL4

Finding Windows
The procedures shown in the sections that follow can be found in the
MWindows module of the API Examples.xls workbook.
To work with a window, we first need to find its handle. In Excel 2002,
the hWnd property was added to the Application object, giving us the handle of the main Excel application window. In previous versions and for all
other top-level windows (that is, windows that are direct children of the
desktop), we can use the FindWindow API call, which is defined as follows:
Declare Function FindWindow Lib "user32" Alias "FindWindowA" _
(ByVal lpClassName As String, _
ByVal lpWindowName As String) As Long

To use the FindWindow function, we need to supply a class name
and/or a window caption. We can use the special constant vbNullString for
either, which tells the function to match on any class or caption. The function searches through all the immediate children of the desktop window
(known as top-level windows), looking for any that have the given class
and/or caption that we specified. To find the main Excel window in versions prior to Excel 2002, we might use the following:
hWndExcel = FindWindow("XLMAIN", Application.Caption)

ANSI vs. Unicode and the Alias Clause
You might have noticed that the declaration for FindWindow contains an
extra clause that we haven’t used before—the Alias clause. All Windows
API functions that have textual parameters come in two flavors: Those that
operate on ANSI strings have an A suffix, whereas those that operate on
Unicode strings have a W suffix. So while all the documentation and
searches on MSDN talk about FindWindow, the Windows DLLs do not
actually contain a function of that name—they contain two functions called
FindWindowA and FindWindowW. We use the Alias statement to provide
the actual name (case sensitive) for the function contained in the DLL. In
fact, as long as we provide the correct name in the Alias clause, we can give
it any name we like:
Declare Function Foo Lib "user32" Alias "FindWindowA" _
(ByVal lpClassName As String, _

ByVal lpWindowName As String) As Long
ApphWnd = Foo("XLMAIN", Application.Caption)

Although VBA stores strings internally as Unicode, it always converts
them to ANSI when passing them to API functions. This is usually sufficient, and it is quite rare to find examples of VB or VBA calling the
Unicode versions. In some cases, however, we need to support the full
Unicode character set and can work around VBA’s conversion behavior by
calling the W version of the API function and using StrConv to do an extra
ANSI-to-Unicode conversion within our API function calls:
Declare Function FindWindow Lib "user32" Alias "FindWindowW" _
(ByVal lpClassName As String, _
ByVal lpWindowName As String) As Long
ApphWnd = FindWindow(StrConv("XLMAIN", vbUnicode), _
StrConv(Application.Caption, vbUnicode))

Finding Related Windows
The problem with the (very common) usage of FindWindow to get the
main Excel window handle is that if we have multiple instances of Excel
open that have the same caption, there is no easy way to tell which one we
get, so we might end up modifying the wrong instance! It is a common
problem if the user typically doesn’t have his workbook windows maximized, because all instances of Excel will then have the same caption of
“Microsoft Excel.”
A more robust and foolproof method is to use the FindWindowEx
function to scan through all children of the desktop window, stopping
when we find one that belongs to the same process as our current instance
of Excel. FindWindowEx works in exactly the same way as FindWindow,
but we provide the parent window handle and the handle of a child
window to start searching after (or zero to start with the first). Listing
9-4 shows a specific ApphWnd function, which calls a generic
FindOurWindow function, which uses the following API functions:
■

GetCurrentProcessID to retrieve the ID of the instance of Excel
running the code

GetDesktopWindow to get the handle of the desktop window, that
we pass to FindWindowEx to look through its children (because all
application windows are children of the desktop)
FindWindowEx to find the next window that matches the given class
and caption
GetWindowThreadProcessID to retrieve the ID of the instance of
Excel that owns the window that FindWindowEx found

Listing 9-4 Foolproof Way to Find the Excel Main Window Handle
'Get the handle of the desktop window
Declare Function GetDesktopWindow Lib "user32" () As Long
'Find a child window with a given class name and caption
Declare Function FindWindowEx Lib "user32" _
Alias "FindWindowExA" _
(ByVal hWnd1 As Long, ByVal hWnd2 As Long, _
ByVal lpsz1 As String, ByVal lpsz2 As String) _
As Long
'Get the process ID of this instance of Excel
Declare Function GetCurrentProcessId Lib "kernel32" () _
As Long
'Get the ID of the process that a window belongs to
Declare Function GetWindowThreadProcessId Lib "user32" _
(ByVal hWnd As Long, ByRef lpdwProcessId As Long) _
As Long
'Foolproof way to find the main Excel window handle
Function ApphWnd() As Long
'Excel 2002 and above have a property for the hWnd
If Val(Application.Version) >= 10 Then
ApphWnd = Application.hWnd
Else
ApphWnd = FindOurWindow("XLMAIN", Application.Caption)
End If
End Function

'Finds a top-level window of the given class and caption
'that belongs to this instance of Excel, by matching the
'process IDs
Function FindOurWindow( _
Optional sClass As String = vbNullString, _
Optional sCaption As String = vbNullString)
Dim
Dim
Dim
Dim

hWndDesktop As Long
hWnd As Long
hProcThis As Long
hProcWindow As Long

'Get the ID of this instance of Excel, to match to
hProcThis = GetCurrentProcessId
'All top-level windows are children of the desktop,
'so get that handle first
hWndDesktop = GetDesktopWindow
Do
'Find the next child window of the desktop that
'matches the given window class and/or caption.
'The first time in, hWnd will be zero, so we'll get
'the first matching window. Each call will pass the
'handle of the window we found the last time,
'thereby getting the next one (if any)
hWnd = FindWindowEx(hWndDesktop, hWnd, sClass, _
sCaption)
'Get the ID of the process that owns the window
GetWindowThreadProcessId hWnd, hProcWindow
'Loop until the window's process matches this process,
'or we didn't find a window
Loop Until hProcWindow = hProcThis Or hWnd = 0
'Return the handle we found
FindOurWindow = hWnd
End Function

The FindOurWindow function can also be used to safely find any of
the top-level windows that Excel creates, such as userforms.

After we’ve found Excel’s main window handle, we can use the
FindWindowEx function to navigate through Excel’s window hierarchy.
Listing 9-5 shows a function to return the handle of a given Excel workbook’s window. To get the window handle, we start at Excel’s main window,
find the desktop (class XLDESK) and then find the window (class
EXCEL7) with the appropriate caption.
Listing 9-5 Function to Find a Workbook’s Window Handle
Private Declare Function FindWindowEx Lib "user32" _
Alias "FindWindowExA" _
(ByVal hWnd1 As Long, ByVal hWnd2 As Long, _
ByVal lpsz1 As String, ByVal lpsz2 As String) _
As Long
'Function to find the handle of a given workbook window
Function WorkbookWindowhWnd(wndWindow As Window) As Long
Dim hWndExcel As Long
Dim hWndDesk As Long
'Get the main Excel window
hWndExcel = ApphWnd
'Find the desktop
hWndDesk = FindWindowEx(hWndExcel, 0, _
"XLDESK", vbNullString)
'Find the workbook window
WorkbookWindowhWnd = FindWindowEx(hWndDesk, 0, _
"EXCEL7", wndWindow.Caption)
End Function

Windows Messages
At the lowest level, windows communicate with each other and with the
operating system by sending simple messages. Every window has a main
message-handling procedure (commonly called its wndproc) to which
messages are sent. Every message consists of four elements: the handle of

the window to which the message is being sent, a message ID and two
numbers that provide extra information about the message (if required).
Within each wndproc, there is a huge case statement that works out what
to do for each message ID. For example, the system will send the
WM_PAINT message to a window when it requires the window to redraw
its contents.
It will probably come as no surprise that we can also send messages
directly to individual windows, using the SendMessage function. The easiest way to find which messages can be sent to which window class is to
search the MSDN library using a known constant and then look in the See
Also list for a link to a list of related messages. Look down the list for a message that looks interesting, then go to its details page to see the parameters
it requires. For example, if we look again at Figure 9-1, we can see that the
EXCEL; window contains a combo box. This combo box is actually
the Name drop-down to the left of the formula bar. Searching the MSDN
library (using Google) with the search term “combo box messages” gives
us a number of relevant hits. One of them takes us to
msdn.microsoft.com/library/en-us/shellcc/platform/
commctls/comboboxes/comboboxes.asp. Looking down the list of

messages we find the CB_SETDROPPEDWIDTH message that we can
use to change the width of the drop-down portion of the Name box. In
Listing 9-6, we use the SendMessage function to make the Name dropdown 200 pixels wide, enabling us to see the full text of lengthy defined
names.
Listing 9-6 Changing the Width of the Name Drop-Down List
Private Declare Function FindWindowEx Lib "user32" _
Alias "FindWindowExA" _
(ByVal hWnd1 As Long, ByVal hWnd2 As Long, _
ByVal lpsz1 As String, ByVal lpsz2 As String) _
As Long
Private Declare Function SendMessage Lib "user32" _
Alias "SendMessageA" _
(ByVal hwnd As Long, ByVal wMsg As Long, _
ByVal wParam As Long, Byval lParam As Long) _
As Long
'Not included in win32api.txt, but found in winuser.h

Private Const CB_SETDROPPEDWIDTH As Long = &H160&
'Make the Name dropdown list 200 pixels wide
Sub SetNameDropdownWidth()
Dim hWndExcel As Long
Dim hWndFormulaBar As Long
Dim hWndNameCombo As Long
'Get the main Excel window
hWndExcel = ApphWnd
'Get the handle for the formula bar window
hWndFormulaBar = FindWindowEx(hWndExcel, 0, _
"EXCEL;", vbNullString)
'Get the handle for the Name combobox
hWndNameCombo = FindWindowEx(hWndFormulaBar, 0, _
"combobox", vbNullString)
'Set the dropdown list to be 200 pixels wide
SendMessage hWndNameCombo, CB_SETDROPPEDWIDTH, 200, 0
End Sub

Changing the Window Icon
When creating a dictator application, the intent is usually to make it look
as though it is a normal Windows application and not necessarily running
within Excel. Two of the giveaways are the application and worksheet
icons. These can be changed to our own icons using API functions. We first
use the ExtractIcon function to get a handle to an icon from a file, then
send that icon handle to the window in a WM_SETICON message, as
shown in Listing 9-7. The SetIcon routine is given a window handle and
the path to an icon file, so it can be used to set either the application’s icon
or a workbook window’s icon. For best use, the icon file should contain
both 32×32 and 16×16 pixel versions of the icon image. Note that when
setting the workbook window’s icon, Excel doesn’t refresh the image to the
left of the menu bar until a window is maximized or minimized/restored,
so you may need to toggle the WindowState to force the update.

Listing 9-7 Setting a Window’s Icon
Private Declare Function ExtractIcon Lib "shell32.dll" _
Alias "ExtractIconA" _
(ByVal hInst As Long, _
ByVal lpszExeFileName As String, _
ByVal nIconIndex As Long) As Long
Private Declare Function SendMessage Lib "user32" _
Alias "SendMessageA" _
(ByVal hwnd As Long, ByVal wMsg As Long, _
ByVal wParam As Long, Byval lParam As Long) _
As Long
Private Const WM_SETICON As Long = &H80
'Set a window's icon
Sub SetIcon(ByVal hWnd As Long, ByVal sIcon As String)
Dim hIcon As Long
'Get the icon handle
hIcon = ExtractIcon(0, sIcon, 0)
'Set the big (32x32) and small (16x16) icons
SendMessage hWnd, WM_SETICON, 1, hIcon
SendMessage hWnd, WM_SETICON, 0, hIcon
End Sub

Changing Windows Styles
If you look at all the windows on your screen, you might notice that they all
look a little different. Some have a title bar, some have minimize and maximize buttons, some have an [x] to close them, some have a 3D look, some
are resizable, some are a fixed size and so on. All of these things are individual attributes of the window and are stored as part of the window’s data
structure. They’re all on/off flags stored as bits in two Long numbers. We
can use the GetWindowLong function to retrieve a window’s style settings,
switch individual bits on or off and write them back using SetWindowLong.
Modifying windows styles in this way is most often done for userforms and
is covered in Chapter 10 — Userform Design and Best Practices.

09_bullen.qxd

1/7/05

274

10:00 AM

Chapter 9

Page 274

Understanding and Using Windows API Calls

Working with the Keyboard
The behavior of many of Excel’s toolbar buttons and some of the dialog
buttons changes if the Shift key is held down when the button is clicked.
For example, the Increase decimal toolbar button normally increases the
number of decimal places shown in a cell, but decreases the number of
decimal places if it is clicked with the Shift key held down. Similarly, when
closing Excel, if you hold down the Shift key when clicking the No button
on the Save Changes? dialog, it acts like a “No to All” button. We can do
exactly the same in our applications by using API functions to examine the
state of the keyboard. The procedures included in this section can be
found in the MKeyboard module of the API Examples.xls workbook.

Checking for Shift, Ctrl, Alt, Caps Lock, Num Lock and
Scroll Lock
The GetKeyState API function tells us whether a given key on the keyboard is currently held down or “on” (in the case of Caps Lock, Num Lock
and Scroll Lock). The function is used by passing a code representing the
key we’re interested in and returns whether the key is being held down or
is “on.” Listing 9-8 shows a function to determine whether one of the six
“special” keys is currently pressed. Note that we have again encapsulated
the key code constants inside a more meaningful enumeration.
Listing 9-8 Checking Whether a Key Is Held Down
Private Declare Function GetKeyState Lib "user32" _
(ByVal vKey As Long) As Integer
Private
Private
Private
Private
Private
Private

Const
Const
Const
Const
Const
Const

VK_SHIFT As Long = &H10
VK_CONTROL As Long = &H11
VK_MENU As Long = &H12
VK_CAPITAL = &H14
VK_NUMLOCK = &H90
VK_SCROLL = &H91

Public Enum GetKeyStateKeyboardCodes
gksKeyboardShift = VK_SHIFT
gksKeyboardCtrl = VK_CONTROL
gksKeyboardAlt = VK_MENU

gksKeyboardCapsLock = VK_CAPITAL
gksKeyboardNumLock = VK_NUMLOCK
gksKeyboardScrollLock = VK_SCROLL
End Enum
Public Function IsKeyPressed _
(ByVal lKey As GetKeyStateKeyboardCodes) As Boolean
Dim iResult As Integer
iResult = GetKeyState(lKey)
Select Case lKey
Case gksKeyboardCapsLock, gksKeyboardNumLock, _
gksKeyboardScrollLock
'For the three 'toggle' keys, the 1st bit says if it's
'on or off, so clear any other bits that might be set,
'using a binary AND
iResult = iResult And 1
Case Else
'For the other keys, the 16th bit says if it's down or
'up, so clear any other bits that might be set, using a
'binary AND
iResult = iResult And &H8000
End Select
IsKeyPressed = (iResult <> 0)
End Function

Bit Masks
The value obtained from the call to GetKeyState should not be interpreted as a simple number, but as its binary representation where each individual bit represents whether a particular attribute is on or off. This is one
of the few functions that return a 16-bit Integer value, rather than the
more common 32-bit Long. The MSDN documentation for GetKeyState
says that “If the high-order bit is 1, the key is down, otherwise the key is
up. If the low-order bit is 1, the key is on, otherwise the key is off.” The

first sentence is applicable for all keys (down/up), whereas the second is
only applicable to the Caps Lock, Num Lock and Scroll Lock keys. It is
possible for both bits to be set, if the Caps Lock key is held down and “on.”
The low-order bit is the rightmost bit, and the high-order bit is the leftmost
(16th) bit. To examine whether a specific bit has been set, we have to apply
a bit mask, to zero-out the bits we’re not interested in, by performing a
binary AND between the return value and a binary value that has a single
1 in the position we’re interested in. In the first case, we’re checking for a
1 in the first bit, which is the number 1. In the second case, we’re checking for a 1 in the 16th bit, i.e. the binary number 1000 0000 0000 0000,
which is easiest to represent in code as the hexadecimal number &h8000.
After we’ve isolated that bit, a zero value means off/up and a nonzero value
means on/down.

Testing for a Key Press
As mentioned previously, at the lowest level, windows communicate
through messages sent to their wndproc procedure. When an application
is busy (such as Excel running some code), the wndproc only processes
critical messages (such as the system shutting down). All other messages
get placed in a queue and are processed when the application next has
some spare time. This is why using SendKeys is so unreliable; it’s not until
the code stops running (or issues a DoEvents statement) that Excel checks
its message queue to see whether there are any key presses to process.
We can use Excel’s message queuing to allow the user to interrupt our
code by pressing a key. Normally, if we want to allow the user to stop a
lengthy looping process, we can either show a modeless dialog with a
Cancel button (as explained in Chapter 10 — Userform Design and Best
Practices), or allow the user to press the Cancel key to jump into the routine’s error handler (as explained in Chapter 12 — VBA Error Handling).
An easier way is to check Excel’s message queue during each iteration of
the loop to see whether the user has pressed a key. This is achieved using
the PeekMessage API function:
Declare Function PeekMessage Lib "user32" _
Alias "PeekMessageA" _
(ByRef lpMsg As MSG, _
ByVal hWnd As Long, _
ByVal wMsgFilterMin As Long, _
ByVal wMsgFilterMax As Long, _
ByVal wRemoveMsg As Long) As Long

Structures
If you look at the first parameter of the PeekMessage function, you’ll see
it is declared As MSG and is passed ByRef. MSG is a windows structure
and is implemented in VBA as a user-defined type. To use it in this case,
we declare a variable of that type and pass it in to the function. The function sets the value of each element of the UDT, which we then read. Many
API functions use structures as a convenient way of passing large amounts
of information into the function, instead of having a long list of parameters.
Many messages that we send using the SendMessage function require a
structure to be passed as the final parameter (as opposed to a single Long
value). In those cases, we use a different form of the SendMessage declaration, where the final parameter is declared As Any and is passed ByRef:
Declare Function SendMessageAny Lib "user32" _
Alias "SendMessageA" _
(ByVal hwnd As Long, ByVal wMsg As Long, _
ByVal wParam As Long, _
ByRef lParam As Any) As Long

When we use this declaration, we’re actually sending a pointer to the
memory where our UDT is stored. If we have an error in the definition of
our UDT, or if we use this version of the declaration to send a message that
is not expecting a memory pointer, the call will at best fail and possibly
crash Excel.
Listing 9-9 shows the full code to check for a key press.
Listing 9-9 Testing for a Key Press
'Type to hold the coordinates of the mouse pointer
Private Type POINTAPI
x As Long
y As Long
End Type
'Type to hold the Windows message information
Private Type MSG
hWnd As Long
'the window handle of the app
message As Long
'the type of message (e.g. keydown)
wParam As Long
'the key code
lParam As Long
'not used
time As Long
'time when message posted

09_bullen.qxd

1/7/05

278

10:00 AM

Chapter 9

pt As POINTAPI
End Type

Page 278

Understanding and Using Windows API Calls

'coordinate of mouse pointer

'Look in the message buffer for a message
Private Declare Function PeekMessage Lib "user32" _
Alias "PeekMessageA" _
(ByRef lpMsg As MSG, ByVal hWnd As Long, _
ByVal wMsgFilterMin As Long, _
ByVal wMsgFilterMax As Long, _
ByVal wRemoveMsg As Long) As Long
'Translate the message from a key code to a ASCII code
Private Declare Function TranslateMessage Lib "user32" _
(ByRef lpMsg As MSG) As Long
'Windows API constants
Private Const WM_CHAR As Long = &H102
Private Const WM_KEYDOWN As Long = &H100
Private Const PM_REMOVE As Long = &H1
Private Const PM_NOYIELD As Long = &H2
'Check for a key press
Public Function CheckKeyboardBuffer() As String
'Dimension variables
Dim msgMessage As MSG
Dim hWnd As Long
Dim lResult As Long
'Get the window handle of this application
hWnd = ApphWnd
'See if there are any "Key down" messages
lResult = PeekMessage(msgMessage, hWnd, WM_KEYDOWN, _
WM_KEYDOWN, PM_REMOVE + PM_NOYIELD)
'If so ...
If lResult <> 0 Then
'... translate the key-down code to a character code,
'which gets put back in the message queue as a WM_CHAR
'message ...
lResult = TranslateMessage(msgMessage)

'... and retrieve that WM_CHAR message
lResult = PeekMessage(msgMessage, hWnd, WM_CHAR, _
WM_CHAR, PM_REMOVE + PM_NOYIELD)
'Return the character of the key pressed,
'ignoring shift and control characters
CheckKeyboardBuffer = Chr$(msgMessage.wParam)
End If
End Function

When we press a key on the keyboard, the active window is sent a
WM_KEYDOWN message, with a low-level code to identify the physical
key pressed. The first thing we need to do, then, is to use PeekMessage to
look in the message queue to see whether there are any pending
WM_KEYDOWN messages, removing it from the queue if we find one. If
we found one, we have to translate it into a character code using
TranslateMessage, which sends the translated message back to Excel’s
message queue as a WM_CHAR message. We then look in the message
queue for this WM_CHAR message and return the character pressed.

Working with the File System and Network
The procedures included in this section can be found in the MFileSys
module of the API Examples.xls workbook.

Finding the User ID
Excel has its own user name property, but does not tell us the user’s network logon ID. This ID is often required in Excel applications for security validation, auditing, logging change history and so on. It can be retrieved
using the API call shown in Listing 9-10.
Listing 9-10 Reading the User’s Login ID
Private Declare Function GetUserName Lib "advapi32.dll" _
Alias "GetUserNameA" _
(ByVal lpBuffer As String, _
ByRef nSize As Long) As Long

'Get the user's login ID
Function UserName() As String
'A buffer that the API function fills with the login name
Dim sBuffer As String * 255
'Variable to hold the length of the buffer
Dim lStringLength As Long
'Initialize to the length of the string buffer
lStringLength = Len(sBuffer)
'Call the API function, which fills the buffer
'and updates lStringLength with the length of the login ID,
'including a terminating null - vbNullChar - character
GetUserName sBuffer, lStringLength
If lStringLength > 0 Then
'Return the login id, stripping off the final vbNullChar
UserName = Left$(sBuffer, lStringLength - 1)
End If
End Function

Buffers
Every API function that returns textual information, such as the user
name, does so by using a buffer that we provide. A buffer comprises a
String variable initialized to a fixed size and a Long variable to tell the function how big the buffer is. When the function is called, it writes the text to
the buffer (including a final Null character) and (usually) updates the
length variable with the number of characters written. (Some functions
return the text length as the function’s result instead of updating the variable.) We can then look in the buffer for the required text. Note that VBA
stores strings in a very different way than the API functions expect, so
whenever we pass strings to API functions, VBA does some conversion for
us behind the scenes. For this to work properly, we always pass strings by
value (ByVal) to API functions, even when the function updates the string.
Some people prefer to ignore the buffer length information, looking
instead for the first vbNullChar character in the buffer and assuming that’s
the end of the retrieved string, so you may encounter usage like that shown
in Listing 9-11.

Listing 9-11 Using a Buffer, Ignoring the Buffer Length Variable
'Get the user's login ID, without using the buffer length
Function UserName2() As String
Dim sBuffer As String * 255
GetUserName sBuffer, 255
UserName2 = Left$(sBuffer, InStr(sBuffer, vbNullChar) - 1)
End Function

Changing to a UNC Path
VBA’s intrinsic ChDrive and ChDir statements can be used to change the
active path prior to using Application.GetOpenFilename, such that
the dialog opens with the correct path preselected. Unfortunately, that can
only be used to change the active path to local folders or network folders
that have been mapped to a drive letter. Note that once set, the VBA
CurDir function will return a UNC path. We need to use API functions to
change the folder to a network path of the form \\server\share\path, as
shown in Listing 9-12. In practice, the SetCurDir API function is one of
the few that can be called directly from your code.
Listing 9-12 Changing to a UNC Path
Private Declare Function SetCurDir Lib "kernel32" _
Alias "SetCurrentDirectoryA" _
(ByVal lpPathName As String) As Long
'Change to a UNC Directory
Sub ChDirUNC(ByVal sPath As String)
Dim lReturn As Long
'Call the API function to set the current directory
lReturn = SetCurDir(sPath)
'A zero return value means an error
If lReturn = 0 Then
Err.Raise vbObjectError + 1, "Error setting path."
End If
End Sub

Locating Special Folders
Windows maintains a large number of special folders that relate to either
the current user or the system configuration. When a user is logged in to
Windows with relatively low privileges, such as the basic User account, it
is highly likely that the user will only have full access to his personal folders, such as his My Documents folder. These folders can usually be found
under C:\Documents and Settings\UserName, but could be located anywhere. We can use an API function to give us the correct paths to these
special folders, using the code shown in Listing 9-13. Note that this listing
contains a subset of all the possible folder constants. The full list can be
found by searching MSDN for “CSIDL Values.” The notable exception
from this list is the user’s Temp folder, which can be found by using the
GetTempPath function. Listing 9-13 includes a special case for this folder,
so that it can be obtained through the same function.
Listing 9-13 Locating a Windows Special Folder
Private Declare Function SHGetFolderPath Lib "shell32" _
Alias "SHGetFolderPathA" _
(ByVal hwndOwner As Long, ByVal nFolder As Long, _
ByVal hToken As Long, ByVal dwFlags As Long, _
ByVal pszPath As String) As Long
Private Declare Function GetTempPath Lib "kernel32" _
Alias "GetTempPathA" _
(ByVal nBufferLength As Long, _
ByVal lpBuffer As String) As Long
'More Commonly used CSIDL values.
'For the full list, search MSDN for "CSIDL Values"
Private Const CSIDL_PROGRAMS As Long = &H2
Private Const CSIDL_PERSONAL As Long = &H5
Private Const CSIDL_FAVORITES As Long = &H6
Private Const CSIDL_STARTMENU As Long = &HB
Private Const CSIDL_MYDOCUMENTS As Long = &HC
Private Const CSIDL_MYMUSIC As Long = &HD
Private Const CSIDL_MYVIDEO As Long = &HE
Private Const CSIDL_DESKTOPDIRECTORY As Long = &H10
Private Const CSIDL_APPDATA As Long = &H1A
Private Const CSIDL_LOCAL_APPDATA As Long = &H1C
Private Const CSIDL_INTERNET_CACHE As Long = &H20

Private
Private
Private
Private

Const
Const
Const
Const

CSIDL_WINDOWS As Long = &H24
CSIDL_SYSTEM As Long = &H25
CSIDL_PROGRAM_FILES As Long = &H26
CSIDL_MYPICTURES As Long = &H27

'Constants used in the SHGetFolderPath call
Private Const CSIDL_FLAG_CREATE As Long = &H8000&
Private Const SHGFP_TYPE_CURRENT = 0
Private Const SHGFP_TYPE_DEFAULT = 1
Private Const MAX_PATH = 260
'Public enumeration to give friendly names for the CSIDL values
Public Enum SpecialFolderIDs
sfAppDataRoaming = CSIDL_APPDATA
sfAppDataNonRoaming = CSIDL_LOCAL_APPDATA
sfStartMenu = CSIDL_STARTMENU
sfStartMenuPrograms = CSIDL_PROGRAMS
sfMyDocuments = CSIDL_PERSONAL
sfMyMusic = CSIDL_MYMUSIC
sfMyPictures = CSIDL_MYPICTURES
sfMyVideo = CSIDL_MYVIDEO
sfFavorites = CSIDL_FAVORITES
sfDesktopDir = CSIDL_DESKTOPDIRECTORY
sfInternetCache = CSIDL_INTERNET_CACHE
sfWindows = CSIDL_WINDOWS
sfWindowsSystem = CSIDL_SYSTEM
sfProgramFiles = CSIDL_PROGRAM_FILES
'There is no CSIDL for the temp path,
'so we need to give it a dummy value
'and treat it differently in the function
sfTemporary = &HFF
End Enum
'Get the path for a Windows special folder
Public Function SpecialFolderPath( _
ByVal uFolderID As SpecialFolderIDs) As String
'Create a buffer of the correct size
Dim sBuffer As String * MAX_PATH
Dim lResult As Long
If uFolderID = sfTemporary Then

'Use GetTempPath for the temporary path
lResult = GetTempPath(MAX_PATH, sBuffer)
'The GetTempPath call returns the length and a
'trailing \ which we remove for consistency
SpecialFolderPath = Left$(sBuffer, lResult - 1)
Else
'Call the function, passing the buffer
lResult = SHGetFolderPath(0, _
uFolderID + CSIDL_FLAG_CREATE, 0, _
SHGFP_TYPE_CURRENT, sBuffer)
'The SHGetFolderPath function doesn't give us a
'length, so look for the first vbNullChar
SpecialFolderPath = Left$(sBuffer, _
InStr(sBuffer, vbNullChar) - 1)
End If
End Function

The observant among you might have noticed that we’ve now come
across all three ways in which buffers are filled by API functions:
■
■
■

GetUserName returns the length of the text by modifying the input
parameter.
GetTempPath returns the length of the text as the function’s return
value.
SHGetFolderPath doesn’t return the length at all, so we search for
the first vbNullChar.

Deleting a File to the Recycle Bin
The VBA Kill statement is used to delete a file, but does not send it to the
recycle bin for potential recovery by the user. To send a file to the recycle
bin, we need to use the SHFileOperation function, as shown in Listing 9-14:
Listing 9-14 Deleting a File to the Recycle Bin
'Structure to tell the SHFileOperation function what to do
Private Type SHFILEOPSTRUCT
hwnd As Long

wFunc As Long
pFrom As String
pTo As String
fFlags As Integer
fAnyOperationsAborted As Boolean
hNameMappings As Long
lpszProgressTitle As String
End Type
Private Declare Function SHFileOperation Lib "shell32.dll" _
Alias "SHFileOperationA" _
(ByRef lpFileOp As SHFILEOPSTRUCT) As Long
Private
Private
Private
Private

Const
Const
Const
Const

FO_DELETE = &H3
FOF_SILENT = &H4
FOF_NOCONFIRMATION = &H10
FOF_ALLOWUNDO = &H40

'Delete a file, sending it to the recycle bin
Sub DeleteToRecycleBin(ByVal sFile As String)
Dim uFileOperation As SHFILEOPSTRUCT
Dim lReturn As Long
'Fill the UDT with information about what to do
With FileOperation
.wFunc = FO_DELETE
.pFrom = sFile
.pTo = vbNullChar
.fFlags = FOF_SILENT + FOF_NOCONFIRMATION + _
FOF_ALLOWUNDO
End With
'Pass the UDT to the function
lReturn = SHFileOperation(FileOperation)
If lReturn <> 0 Then
Err.Raise vbObjectError + 1, "Error deleting file."
End If
End Sub

There are two things to note about this function. First, the function
uses a user-defined type to tell it what to do, instead of the more common
method of having multiple input parameters. Second, the function returns
a value of zero to indicate success. If you recall the SetCurDir function in
Listing 9-12, it returns a value of zero to indicate failure! The only way to
know which to expect is to check the Return Values section of the function’s information page on MSDN.

Browsing for a Folder
All versions of Excel have included the GetOpenFilename and
GetSaveAsFilename functions to allow the user to select a filename to
open or save. Excel 2002 introduced the common Office FileDialog object,
which can be used to browse for a folder, using the code shown in Listing
9-15, which results in the dialog shown in Figure 9-3.
Listing 9-15 Using Excel 2002’s FileDialog to Browse for a Folder
'Browse for a folder, using the Excel 2002 FileDialog
Sub BrowseForFolder()
Dim fdBrowser As FileDialog
'Get the File Dialog object
Set fdBrowser = Application.FileDialog(msoFileDialogFolderPicker)
With fdBrowser
'Initialize it
.Title = "Select Folder"
.InitialFileName = "c:\"
'Display the dialog
If .Show Then
MsgBox "You selected " & .SelectedItems(1)
End If
End With
End Sub

Figure 9-3 The Standard Office 2002 Folder Picker Dialog

We consider this layout far too complicated, when all we need is a simple tree view of the folders on the computer. We can use API functions to
show the standard Windows Browse for folder dialog shown in Figure 9-4,
which our users tend to find much easier to use. The Windows dialog also
gives us the option to display some descriptive text to tell our users what
they should be selecting.
Callbacks
So far, every function we’ve encountered just does its thing and returns
its result. However, a range of API functions (including the
SHBrowseForFolder function that we’re about to use) interact with
the calling program while they’re working. This mechanism is known as a
callback. Excel 2000 added a VBA function called AddressOf, which provides the address in memory where a given procedure can be found. This
address is passed to the API function, which calls back to the procedure
found at that address as required. For example, the EnumWindows function iterates through all the top-level windows, calling back to the procedure with the details of each window it finds. Obviously, the procedure
being called must be defined exactly as Windows expects it to be so the
API function can pass it the correct number and type of parameters.

Figure 9-4 The Standard Windows Folder Picker Dialog

The SHBrowseForFolder function uses a callback to tell us when the
dialog is initially shown, enabling us to set its caption and initial selection,
and each time the user selects a folder, enabling us to check the selection
and enable/disable the OK button. The full text for the function is contained in the MBrowseForFolder module of the API Examples.xls workbook and a slightly simplified version is shown in Listing 9-16.
Listing 9-16 Using Callbacks to Interact with the Windows File Picker Dialog
'UDT to pass information to the SHBrowseForFolder function
Private Type BROWSEINFO
hOwner As Long
pidlRoot As Long
pszDisplayName As String
lpszTitle As String
ulFlags As Long
lpfn As Long
lParam As Long
iImage As Long
End Type

'Commonly used ulFlags constants
'Only return file system directories.
'If the user selects folders that are not
'part of the file system (such as 'My Computer'),
'the OK button is grayed.
Private Const BIF_RETURNONLYFSDIRS As Long = &H1
'Use a newer dialog style, which gives a richer experience
Private Const BIF_NEWDIALOGSTYLE As Long = &H40
'Hide the default 'Make New Folder' button
Private Const BIF_NONEWFOLDERBUTTON As Long = &H200

'Messages sent from dialog to callback function
Private Const BFFM_INITIALIZED = 1
Private Const BFFM_SELCHANGED = 2

'Messages sent to browser from callback function
Private Const WM_USER = &H400
'Set the selected path
Private Const BFFM_SETSELECTIONA = WM_USER + 102
'Enable/disable the OK button
Private Const BFFM_ENABLEOK = WM_USER + 101

'The maximum allowed path
Private Const MAX_PATH = 260
'Main Browse for directory function
Declare Function SHBrowseForFolder Lib "shell32.dll" _
Alias "SHBrowseForFolderA" _
(ByRef lpBrowseInfo As BROWSEINFO) As Long
'Gets a path from a pidl
Declare Function SHGetPathFromIDList Lib "shell32.dll" _
Alias "SHGetPathFromIDListA" _
(ByVal pidl As Long, _
ByVal pszPath As String) As Long

'Used to set the browse dialog's title
Declare Function SetWindowText Lib "user32" _
Alias "SetWindowTextA" _
(ByVal hwnd As Long, _
ByVal lpString As String) As Long
'A versions of SendMessage, to send strings to the browser
Private Declare Function SendMessageString Lib "user32" _
Alias "SendMessageA" (ByVal hwnd As Long, _
ByVal wMsg As Long, ByVal wParam As Long, _
ByVal lParam As String) As Long
'Variables to hold the initial options,
'set in the callback function
Dim msInitialPath As String
Dim msTitleBarText As String
'The main function to initialize and show the dialog
Function GetDirectory(Optional ByVal sInitDir As String, _
Optional ByVal sTitle As String, _
Optional ByVal sMessage As String, _
Optional ByVal hwndOwner As Long, _
Optional ByVal bAllowCreateFolder As Boolean) _
As String
'A variable to hold the UDT
Dim uInfo As BROWSEINFO
Dim sPath As String
Dim lResult As Long
'Check that the initial directory exists
On Error Resume Next
sPath = Dir(sInitDir & "\*.*", vbNormal + vbDirectory)
If Len(sPath) = 0 Or Err.Number <> 0 Then sInitDir = ""
On Error GoTo 0
'Store the initials setting in module-level variables,
'for use in the callback function
msInitialPath = sInitDir
msTitleBarText = sTitle
'If no owner window given, use the Excel window

'N.B. Uses the ApphWnd function in MWindows
If hwndOwner = 0 Then hwndOwner = ApphWnd
'Initialise the structure to pass to the API function
With uInfo
.hOwner = hwndOwner
.pszDisplayName = String$(MAX_PATH, vbNullChar)
.lpszTitle = sMessage
.ulFlags = BIF_RETURNONLYFSDIRS + BIF_NEWDIALOGSTYLE _
+ IIf(bAllowCreateFolder, 0, BIF_NONEWFOLDERBUTTON)
'Pass the address of the callback function in the UDT
.lpfn = LongToLong(AddressOf BrowseCallBack)
End With
'Display the dialog, returning the ID of the selection
lResult = SHBrowseForFolder(uInfo)
'Get the path string from the ID
GetDirectory = GetPathFromID(lResult)
End Function

'Windows calls this function when the dialog events occur
Private Function BrowseCallBack (ByVal hwnd As Long, _
ByVal Msg As Long, ByVal lParam As Long, _
ByVal pData As Long) As Long
Dim sPath As String
'This is called by Windows, so don't allow any errors!
On Error Resume Next
Select Case Msg
Case BFFM_INITIALIZED
'Dialog is being initialized,
'so set the initial parameters
'The dialog caption
If msTitleBarText <> "" Then
SetWindowText hwnd, msTitleBarText
End If

'The initial path to display
If msInitialPath <> "" Then
SendMessageString hwnd, BFFM_SETSELECTIONA, 1, _
msInitialPath
End If
Case BFFM_SELCHANGED
'User selected a folder
'lParam contains the pidl of the folder, which can be
'converted to the path using GetPathFromID
'sPath = GetPathFromID(lParam)
'We could put extra checks in here,
'e.g. to check if the folder contains any workbooks,
'and send the BFFM_ENABLEOK message to enable/disable
'the OK button:
'SendMessage hwnd, BFFM_ENABLEOK, 0, True/False
End Select
End Function

'Converts a PIDL to a path string
Private Function GetPathFromID(ByVal lID As Long) As String
Dim lResult As Long
Dim sPath As String * MAX_PATH
lResult = SHGetPathFromIDList(lID, sPath)
If lResult <> 0 Then
GetPathFromID = Left$(sPath, InStr(sPath, Chr$(0)) - 1)
End If
End Function
'VBA doesn't let us assign the result of AddressOf
'to a variable, but does allow us to pass it to a function.
'This 'do nothing' function works around that problem
Private Function LongToLong(ByVal lAddr As Long) As Long
LongToLong = lAddr
End Function

09_bullen.qxd

1/7/05

10:00 AM

Page 293

Working with the File System and Network

293

Let’s take a closer look at how this all works. First, most of the shell
functions use things called PIDLs to uniquely identify folders and files.
For simplicity’s sake, you can think of a PIDL as a handle to a file or folder,
and there are API functions to convert between the PIDL and the normal
file or folder name.
The GetDirectory function is the main function in the module and is
the function that should be called to display the dialog. It starts by validating the (optional) input parameters, then populates the BROWSEINFO
user-defined type that is used to pass all the required information to the
SHBrowseForFolder function. The hOwner element of the UDT is used
to provide the parent window for the dialog, which should be the handle
of the main Excel window, or the handle of the userform window if showing this dialog from a userform. The ulFlags element is used to specify
detailed behavior for the dialog, such as whether to show a Make Folder
button. The full list of possible flags and their purpose can be found on
MSDN by searching for the SHBrowseForFolder function. The lpfn
element is where we pass the address of the callback function,
BrowseCallBack. We have to wrap the AddressOf value in a simple
LongToLong function, because VB doesn’t let us assign the value directly
to an element of a UDT.
After the UDT has been initialized, we pass it to the
SHBrowseForFolder API function. That function displays the dialog and
Windows calls back to our BrowseCallBack function, passing the
BFFM_INITIALIZED message. We respond to that message by setting
the dialog’s caption (using the SetWindowText API function) and the initial folder selection (by sending the BFFM_SETSELECTIONA message
back to the dialog with the path string).
Every time the user clicks a folder, it triggers a Windows callback to
our BrowseCallBack function, passing the BFFM_SELCHANGED message and the ID of the selected folder. All the code to respond to that message is commented out in this example, but we could add code to check
whether the folder is a valid selection for our application (such as whether
it contains any workbooks) and enable/disable the OK button appropriately (by sending the BFFM_ENABLEOK message back to the dialog).
When the user clicks the OK or Cancel button, the function returns
the ID of the selected folder and execution continues back in the
GetDirectory function. We get the textual path from the returned ID and
return it to the calling code.

Practical Examples
All the routines included in this chapter have been taken out of actual
Excel applications, so are themselves practical examples of API calls.
The PETRAS application files for this chapter can be found on the CD
in the folder \Application\Ch09—Understanding and Using Windows API
Calls and now includes the following files:
■
■
■
■
■
■

PetrasTemplate.xlt—The timesheet template
PetrasAddin.xla—The timesheet data-entry support add-in
PetrasReporting.xla—The main reporting application
PetrasConsolidation.xlt—A template to use for new results
workbooks
Debug.ini—A dummy file that tells the application to run in debug
mode
PetrasIcon.ico—A new icon file, to use for Excel’s main window

PETRAS Timesheet
Until this chapter, the location used by the Post to Network routine has
used Application.GetOpenFilename to allow the user to select the directory to save the timesheet workbook to. The problem with that call is that
the directory must already contain at least one file. In this chapter, we add
the BrowseForFolder dialog and use that instead of GetOpenFilename,
which allows empty folders to be selected.
We’ve also added a new feature to the timesheet add-in. In previous
versions you were prompted to specify the consolidation location the first
time you posted a timesheet workbook to the network. When you selected
a location, that location was stored in the registry and from there on out
the application simply read the location from the registry whenever you
posted a new timesheet.
What this didn’t take into account is the possibility that the consolidation location might change. If it did, you would have no way, short of
editing the application’s registry entries directly, of switching to the new
location. Our new Specify Consolidation Folder feature enables you to
click a button on the toolbar and use the Windows browse for folders

dialog to modify the consolidation folder. The SpecifyConsolidationFolder
procedure is shown in Listing 9-17 and the updated toolbar is shown in
Figure 9-5.
Listing 9-17 The New SpecifyConsolidationFolder Procedure
Public Sub SpecifyConsolidationFolder()
Dim sSavePath As String
InitGlobals
' Get the current consolidation path.
sSavePath = GetSetting(gsREG_APP, gsREG_SECTION, _
gsREG_KEY, "")
' Display the browse for folders dialog with the initial
' path display set to the current consolidation folder.
sSavePath = GetDirectory(sSavePath, _
gsCAPTION_SELECT_FOLDER, gsMSG_SELECT_FOLDER)
If Len(sSavePath) > 0 Then
' Save the selected path to the registry.
If Right$(sSavePath, 1) <> "\" Then _
sSavePath = sSavePath & "\"
SaveSetting gsREG_APP, gsREG_SECTION, _
gsREG_KEY, sSavePath
End If
End Sub

Table 9-2 summarizes the changes that have been made to the
timesheet add-in for this chapter.

Figure 9-5 The Updated PETRAS Timesheet Toolbar

Table 9-2 Changes to the PETRAS Timesheet Add-in to Use the BrowseForFolder
Routine
Module

Procedure

MBrowseForFolder
(new module)
MEntryPoints

PostTimeEntriesToNetwork

SpecifyConsolidationFolder

Change
Included the entire
MBrowseForFolder module
shown in Listing 9-16
Added call to the
GetDirectory function in
MBrowseForFolder
New feature to update the
consolidation folder location

PETRAS Reporting
The changes made to the central reporting application for this chapter are
to display a custom icon for the application and to enable the user to close
all the results workbooks simultaneously, by holding down the Shift key
while clicking the File > Close menu. The detailed changes are shown in
Table 9-3, and Listing 9-18 shows the new MenuFileClose routine that
includes the check for the Shift key.
Table 9-3 Changes to the PETRAS Reporting Application for Chapter 9
Module

Procedure

Change

MAPIWrappers
(new module)
MAPIWrappers
(new module)

ApphWnd

MAPIWrappers

IsKeyPressed

Included Listing 9-4 to obtain the
handle of Excel’s main window
Included Listing 9-7 to display
a custom icon, read from the
new PetrasIcon.ico file.
Included Listing 9-8 to check for
the Shift key held down when
clicking File > Close
Added a constant for the icon
filename
Added a call to SetIcon
Added check for Shift key being
held down, shown in Listing
9-17, doing a Close All if so

SetIcon

MGlobals
MWorkspace
MEntryPoints

ConfigureExcelEnvironment
MenuFileClose

Listing 9-18 The New MenuFileClose Routine, Checking for a Shift+Close
'Handle the File > Close menu
Sub MenuFileClose()
Dim wkbWorkbook As Workbook
'Ch09+
'Check for a Shift+Close
If IsKeyPressed(gksKeyboardShift) Then
'Close all results workbooks
For Each wkbWorkbook In Workbooks
If IsResultsWorkbook(wkbWorkbook) Then
CloseWorkbook wkbWorkbook
End If
Next
Else
'Ch09'Close only the active workbook
If IsResultsWorkbook(ActiveWorkbook) Then
CloseWorkbook ActiveWorkbook
End If
End If
End Sub

Later chapters, particularly Chapter 10 — Userform Design and Best
Practices, use more of the routines and concepts introduced in this
chapter.

Conclusion
The Excel object model provides an extremely rich set of tools for us to use
when creating our applications. By including calls to Windows API functions, we can enhance our applications to give them a truly professional
look and feel.

This chapter has explained most of the uses of API functions that are
commonly encountered in Excel application development. All the fundamental concepts have been explained and you should now be able to interpret and understand new uses of API functions as you encounter them.
All of the example routines included in this chapter have been taken
from actual Excel applications and are ready for you to use in your own
workbooks.

